
Abstract

Data-ow optimizations are usually implemented on low-level intermediate repre-

sentations. This is not appropriate for source-to-source optimizations, which re-

construct a source level program after transformation. In this paper we show how

constant propagation, a well known data-ow optimization problem, can be imple-

mented on abstract syntax trees in Stratego, a rewriting system extended with

programmable rewriting strategies for the control over the application of rules and

dynamic rewrite rules for the propagation of information.

1 Introduction

Optimizing compilers for imperative languages apply data-ow optimizations

to improve the performance of programs [14,2]. Data-ow optimizations such

as constant propagation, copy propagation, and dead code elimination trans-

form or eliminate statements or expressions based on data ow information

that is propagated along the control-ow paths of the program. For example,

in constant propagation the assignment of a constant value to a variable is

propagated to occurrences of the variable, which can then be replaced by the

constant value.

Data-ow optimizations in compilers are usually performed on an interme-

diate representation (IR) in which control is expressed by means of labels and

jumps, and memory access is expressed in terms of registers and memory stores

and fetches [16,14,22], or in terms of stack operations. For the implementation

of optimizations, the at list of instructions is turned into a control-ow graph,

which guides the propagation of information. Implementing optimizations on

a low-level intermediate representation has the advantage that the implemen-

tation is reusable for all source languages that are translated to the IR.

On the other hand, the translation to IR leads to the elimination of infor-

mation which was available at source level. In compilation this leads to over-

head, e.g., control-ow analysis is required to recover control-ow structures

such as conditional branches and loops, which were available in the original

source program. In source-to-source transformation the loss of information

makes it impossible to recreate source programs close to the original.

Source-to-source transformation systems manipulate programs at source

level and produce transformed programs in the same (high-level) language.

Applications include instrumentation, aspect weaving, software renovation,

and domain-speci�c optimization. For example, the CodeBoost transforma-

tion system [4] transforms C++ programs with the purpose of eliminating the

overhead caused by an abstract programming style, and applying optimiza-

tions based on knowledge of the application domain, which the C++ compiler

does not have.

To (re)construct a source level program after transformation, source-to-

source transformations are best applied to a representation that is as close to

the original source as possible | abstract syntax trees. Since transformations

on control-ow graphs are not directly applicable to abstract syntax trees,

source-to-source optimizers cannot reuse optimizations on control-ow graphs.

Term rewriting suggests itself as a natural paradigm for transformations on

(abstract syntax) trees. While optimizations such as algebraic simpli�cation

and constant folding can indeed be expressed elegantly using rewrite rules,

data-ow optimizations such as constant propagation and copy propagation

cannot be expressed directly using rewriting. First of all rewrite rules are

context-free, i.e., can only access local information, while in data-ow opti-

mizations the propagation of context information is required. Furthermore,

standard rewriting strategies apply rules exhaustively throughout the tree.

In data-ow optimizations, the control-ow paths of the program should be

followed.

In this paper we show how data-ow optimizations on abstract syntax

trees can be implemented in Stratego [19], a rewriting system extended with

programmable rewriting strategies [20] for the control over the application of

rules and dynamic rewrite rules [18] for the propagation of information.

To illustrate the ideas we develop the speci�cation of constant propagation

for Tiger [2]|an imperative language with nested functions. In Section 2 we

present the abstract syntax of Tiger expressions and de�ne constant folding

rules and a generic strategy for simpli�cation of Tiger expressions. In Section 3

we add assignment and sequential composition to the language. Constant pro-

pagation for such straight-line programs can be de�ned using dynamic rules,

which take care of substituting constants for the variables to which they have

been assigned. The strategy ensures that the rules are applied in the right

order. In Section 4 we add structured control-ow constructs such as condi-

tionals and loops. Propagation of constant assignments forks at such control

ow statements and needs to be combined at the meeting points. Appropriate

operations support saving, restoring, and computing `meet' operations on sets

of dynamic rules.

The building blocks developed for the speci�cation of constant propaga-

tion can be considered as a framework for the implementation of data-ow

optimizations. We have used these building blocks to implement copy pro-

pagation, common sub-expression elimination and dead code elimination. In

Section 5 we discuss this generalization, compare the approach to related work,

and mention future work.

We assume that the reader is familiar with the basic concepts of term

rewriting. We will explain the Stratego constructs that are used, but not in

depth. For an introduction to the basic concepts of Stratego we refer the

reader to [20,18].

2 Constant Folding

Constant folding is the simpli�cation of operator expressions with known cons-

tant values as operands, i.e., reducing expressions of the form c1 op c2, for

some operator op with constant arguments c1 and c2. For example, the ex-

pression 3 + (6 * (5 - 2)) / 2 can be simpli�ed at compile time to 12,

by applying the laws of arithmetic. Constant folding is the goal of constant

propagation; if a variable can be replaced by its constant value, run-time

computations may be replaced with compile-time computations.

Before we study constant propagation, we consider the speci�cation of

constant folding on expressions. Figure 1 presents the signature of abstract

syntax trees representing Tiger expressions. Expressions consist of arith-

metic and relational operations on integer values, boolean values and strings.

Boolean values in Tiger are represented with integer values, false is repre-

sented with the integer zero and true with any non zero integer value.

module Tiger-Expressions

signature

constructors

Int : IntConst -> Exp

String : String -> Exp

Var : Id -> Var

BinOp : BinOp * Exp * Exp -> Exp

RelOp : RelOp * Exp * Exp -> Exp

Call : Var * List (Exp) -> Exp

NilExp : Exp

module Operators

signature

constructors

PLUS : BinOp MINUS : BinOp MUL : BinOp DIV : BinOp

AND : BinOp OR : BinOp EQ : RelOp NE : RelOp

LT : RelOp LE : RelOp GE : RelOp GT : RelOp

Fig. 1. Abstract syntax tree of Tiger expressions

EvalBinOp: BinOp(PLUS, Int(i), Int(j)) -> Int(<add>(i,j))

EvalBinOp: BinOp(PLUS, Int(0), e) -> e

EvalBinOp: BinOp(PLUS, e, Int(0)) -> e

EvalBinOp: BinOp(MINUS,Int(i),Int(j)) -> Int(<subt>(i,j))

EvalBinOp: BinOp(MINUS, Var(x), Var(x)) -> Int(0)

EvalBinOp: BinOp(MUL, Int(i), Int(j)) -> Int(<mul>(i,j))

EvalBinOp: BinOp(MUL, Var(x), Int(1)) -> Var(x)

EvalBinOp: BinOp(MUL, Int(1), Var(x)) -> Var(x)

EvalBinOp: BinOp(MUL, Int(0), e) -> Int(0)

EvalBinOp: BinOp(MUL, e, Int(0)) -> Int(0)

EvalBinOp: BinOp(DIV, Int(i), Int(j)) -> Int(<div>(i,j))

where <not(0)> j

EvalRelOp: RelOp(EQ, Int(a), Int(b)) -> Int(<eval-rel(eq)>(a,b))

EvalRelOp: RelOp(NE, Int(a), Int(b)) ->

Int(<eval-rel(not(eq))>(a,b))

EvalRelOp: RelOp(GT, Int(a), Int(b)) -> Int(<eval-rel(gt)>(a,b))

EvalRelOp: RelOp(LT, Int(a), Int(b)) -> Int(<eval-rel(lt)>(a,b))

EvalRelOp: RelOp(GE, Int(a), Int(b)) -> Int(<eval-rel(geq)>(a,b))

EvalRelOp: RelOp(LE, Int(a), Int(b)) -> Int(<eval-rel(leq)>(a,b))

Fig. 2. Constant folding rules for Tiger expressions

2.1 Constant Folding Rules

The simpli�cation of operators with constant operands can be expressed di-

rectly using rewrite rules. Figure 2 presents several constant folding rules for

arithmetic and Boolean operators. The rules are labeled such that they can

be used in a strategy.

As part of the construction of a new term on the right-hand side of a

rule, a strategy s can be applied to a subterm t using the notation <s>t.

In the constant folding rules this is used to compute integer operations. For

example, <add>(i,j) denotes a call to the primitive add for addition of the

integer values i and j. The strategy eval-rel(s) applies the parameter

strategy s and turns success into the integer value 1 and failure into 0.

2.2 Constant Folding Strategy

In Stratego rewrite rules are not applied automatically. Instead, the speci�-

cation should determine which rules are applied where and in what order by

de�ning a rewriting strategy. For the application of the constant folding rules

a single, bottom-up pass over the expression tree suÆces to reduce all possible

constant operator applications. This can be speci�ed with the generic term

traversal bottomup, as follows:

constant-folding = bottomup(try(fold))

The fold strategy used in the de�nition of constant-folding is de�ned as

the choice between the EvalBinOp and EvalRelOp rules from Figure 2:

module Tiger-Statements

signature

constructors

Assign : LValue * Exp -> Exp

Let : List (Dec) * List (Exp) -> Exp

VarDec : Id * Option (TypeId) * Exp -> Dec

Seq : List (Exp) -> Exp

Fig. 3. Reduced abstract syntax of Tiger.

fold = EvalBinOp + EvalRelOp

The operator s1 + s2 denotes the choice between the strategies s1 and s2.
Finally, the generic traversal strategy bottomup traverses a tree by �rst ap-
plying itself recursively to all direct subterms of a node using all and then
applying the parameter strategy s:

bottomup(s) = all(bottomup(s)); s

Note that the bottomup(s) strategy is a one pass traversal.

3 Constant propagation

Constant propagation is de�ned as the process to discover values that are

constant on all possible executions of a program and to propagate these values

through the program. This is a common data-ow problem and the results of

this speci�cation can be generalized towards di�erent analyses.

For simplicity, we �rst consider a reduced version of Tiger which consists of

expressions, assignment, let and sequential composition. Figure 3 de�nes this

language. When reduced to this syntax, constant propagation comes down to

propagating constant assignments through a sequence of assignments. It is

important that constants are only substituted for variables in program points

where the constant assignment is valid. For example, in the transformation

a := 3

b := a + 2

a := y

b := a + b

)

a := 3

b := 5

a := y

b := a + 5

the �rst occurrence of a in a right-hand side is replaced by 3, but the second

is not because of the intervening de�nition of a.

3.1 Dynamic Propagation Rules

As the example above illustrates, constant propagation works by replacing a
variable with the constant value which is assigned to it. Since this information
is not available at the place where a variable is used, the association should
be established at the place where the variable is de�ned, i.e., the assignment,
and be available at usage sites. Dynamic rewrite rules [18] were designed

for exactly this purpose. The following assign-cp strategy recognizes an
assignment statement and generates a dynamic rule PropConst which rewrites
the variable Var(x) to the expression e assigned to it:

assign-cp =

?Assign(Var(x), e);

where(<is-value> e

< rules(PropConst: Var(x) -> e)

+ rules(PropConst: Var(x) -> Undefined))

The match construct ?t, matches the term pattern t, just like the left-hand

side of a rewrite rule. The s1 < s2 + s3 construct is a guarded deterministic

choice operator. If the guard strategy s1 succeeds, s2 is applied to the result

of s1, else s3 is applied to the original term. In this case the PropConst rule

is only generated when e is a value, i.e., a constant. In the other case the

dynamic rule is unde�ned.

The rules(L : t1 -> t2) construct generates a new (dynamic) rewrite

rule with label L, which rewrites a term matching t1 to t2, just like an ordinary

rule L : t1 -> t2. The di�erence is that variables in t1 and t2 that are

bound in the context are inherited from that context. Thus, in the case above,

the (meta)variables x and e are instantiated when the rule is generated, and

not when it is applied. The assign-cp strategy can be applied several times

for the same variable while a program is being traversed. Every time that this

strategy is applied a new rule is generated overriding a previous rule for the

same variable. However, rules for di�erent variables are not overriden.

In case the expression e assigned to the variable x is not a constant, it

is necessary to unde�ne the PropConst rule in order to kill a previously

generated rule for the same variable. For instance, in the previous exam-

ple the rule PropConst: Var("a") -> Int(3) is unde�ned by the assign-

ment Assign(Var("a"),Var("y")). At this point, the rule is overriden with

PropConst: Var("a") -> Undefined. The application of an unde�ned rule

always fails. Thus, attempting to apply this rule to the term Var("a") fails.

Rules for di�erent varialbes are not unde�ned.

3.2 Propagation Strategy

The propagation strategy const-prop1 for straight-line code is de�ned in

Figure 4. The strategy is almost the same as the constant-folding stra-

tegy of the previous section. That is, during a bottom-up traversal over the

program, constant folding rules are applied. The combined constant folding

rules fold of the previous section are extended with the dynamically gene-

rated PropConst rules and with the assign-cp strategy for generating these

rules. Thus, if a variable is encountered during constant folding, the dynamic

rule PropConst is applied to discover its constant value and replace it with

that value. If the variable is not associated with a constant, the rule will not

succeed, and the variable remains in the program.

module Tiger-ConstProp

strategies

const-prop1 = bottomupS(try(fold1 + assign-cp), control-flow1)

fold1 = PropConst + fold

control-flow1(cp) = Assign(id, cp)

Fig. 4. Constant propagation for straight line code.

However, the strategy cannot be a complete bottomup traversal. Consider,

for example, the sequence a := 3; a := a + 2. Figure 5 illustrates the visit

order through the abstract syntax tree for this sequence using the generic

bottomup(s) traversal. The numbers on the edges indicate at which point the

parameter strategy s is applied to the node the edge points to. Clearly the

�rst assignment is visited before the variables in the second one, thus correc-

tly propagating the constant 3 to the variable a in the second assignment.

However, the variable a in the left-hand side of the second assignment would

also be replaced. To avoid this the constant propagation traversal should not

visit the left-hand sides of assignments. This is achieved with the bottomupS

strategy, a variant of bottomup, which is de�ned as:

bottomupS(s, skip: a * (a -> a) -> a) =

(skip(bottomupS(s, skip)) <+ all(bottomupS(s, skip))) ; s

The skip parameter provides an alternative for the generic traversal with all.

In Figure 4 the skip strategy is instantiated with the control-flow1 strategy,

which only applies the constant propagation strategy to the right-hand side of

assignments. This is achieved using the congruence strategy Assign(id, cp).

A congruence strategy c(s1,...,sn) matches with a term c(t1,...,tn) and

constructs the term c(<s1>t1,...,<sn>tn), applying the argument strate-

gies to the corresponding argument terms. Thus, Assign(id,cp) applies the

identity strategy id to the left-hand side of an assignment and constant propa-

gation cp to the right-hand side. Thus, achieving a traversal that is faithful to

the control ow of the programming language. In Figure 5, the corresponding

visit order is the one where visits 1 and 4 are skipped.

Seq

Assign

3

Assign

9

Var("a")

(1)

Int(3)

2

Var("a")

(4)

BinOp

8

PLUS

5

Var("a")

6

Int(2)

7

Fig. 5. Visit order for bottomup(s)

4 Propagation and Control-Flow

So far we have considered constant propagation in straight line code. Real pro-

grams use conditionals and loops which fork or iterate the data ow. Control-

ow statements in Tiger consist of the if-then-else and if-then conditional

statements, and the for and while iterative statements. Figure 6 gives the

abstract syntax of these constructs.

The propagation of constant values on control-ow statements forks and

merges according to di�erent execution paths. The propagation of constant

values has to follow the di�erent paths imposed by the data ow of the pro-

gram. For instance, consider the following transformation:

b := 4;

d := 2;

if b > x then

(a := 6 * d;

b := 45)

else

(b := 6;

a := b + b);

c := a + d + b

)

b := 4;

d := 2;

if 4 > x then

(a := 12;

b := 45)

else

(b := 6;

a := 12);

c := 14 + b

where the variables a, b, d are of interest: b and d contain constant values

before the occurrence of the if statement and these values are propagated

to the branches of the if. After the if, a and d contain constant values, a

contains a constant value, because it is de�ned with the same constant value

in both branches. The variable d is not de�ned in the if statement. On the

other hand, b is assigned a di�erent value in each branch, hence it does not

contain a constant value after the if, even though it does contain a constant

value locally in the branches.

4.1 Unreachable Code Elimination

Control-ow statements fork the execution paths of a program. At execution

time some paths are not reachable. To avoid considering unreachable code,

the constant folding rules are extended to eliminate branches from control-

ow statements when conditions can be evaluated statically. Consider the

module Tiger-Statements

signature

constructors

IfThen : Exp * Exp -> Exp

If : Exp * Exp * Exp -> Exp

While : Exp * Exp -> Exp

For : Var * Exp * Exp * Exp -> Exp

Fig. 6. Control-ow statements of Tiger.

ElimIf : If(Int(0), e1, e2) -> e2

ElimIf : If(Int(i), e1, e2) -> e1 where <not(eq)> (i, 0)

ElimIfThen: IfThen(Int(0),e1) -> Seq([])

ElimIfThen: IfThen(Int(i),e1) -> e1

ElimWhile : While(Int(0),e) -> Seq([])

ElimFor : For(v,Int(i),Int(j),e) -> Seq([])

where <eval-rel(lt)> (j,i) => 1

Fig. 7. Rules to eliminate unreachable code

If statement with three argument expressions, a condition, a then, and an

else expressions. Constant propagation �rst evaluates the condition (as the

execution of the program will do). If this evaluation results in a constant

value, it determines which branch will be executed (Tiger uses 0 for false and

another integer value for true). Unreachable code elimination can be expressed

by means of rewrite rules as shown in Figure 7. The application of the ElimIf

and ElimIfThen rules selects the executable branch and the If expression is

replaced by the surviving branch. The search for constants will continue in

the selected branch. If a loop will not be executed it can be removed from the

program.

4.2 Distributing and Merging Propagation Rules

The speci�cation of constant propagation is based on dynamic rules to pro-

pagate constant values. Dynamic propagation rules must follow the data ow

of a program. To achieve this behavior, propagation rules are passed to the

execution paths and merged at the point where execution paths are joined.

Figure 8 depicts the data ow of an if statement. A structured if-then-else

statement forks and merges two execution paths. At the point where the if-

then-else statement forks, dynamic propagation rules are passed to the then

and else branches. At the merge point, two di�erent versions of dynamic

propagation rules are combined.

Propagation rules carry the information required to propagate constants.

These rules have to traverse the program in a way that emulates the data ow

of a program. In order to manipulate active propagation rules, operations

such as save, restore and intersect(merge) are used.

elsethen

continuation
merge point

If fork point

Fig. 8. Data ow of a structured If statement

Input Term Transformation Rules Transf. Term

Active rules Sets of rules

b := 4; PC: Var(b) -> 4 (1) b := 4;

d := 2; PC: Var(d) -> 2 (2) d := 2;

if b>x then saving active rules dr-in ={PC: Var(b) -> 4, if 4>x then

PC: Var(d) -> 2}

(a := 6 * d; PC: Var(a) -> 12 (3) (a := 12;

b := 45 ) PC: Var(b) -> 45 (4) b := 45)

else saving active rules dr-then ={PC: Var(d) -> 2, else

PC: Var(a) -> 12,

PC: Var(b) -> 45}

PC: Var(b) -> 4 (5) restoring dr-in rules

PC: Var(d) -> 2 (6)

(b := 6; PC: Var(b) -> 6 (7) overrides rule (5) (b := 6;

a := b + b); PC: Var(a) -> 12 (8) a := 12);

T
dr-then = intersection with active

{PC: Var(d) -> 2, rules (6,7,8)

PC: Var(a) -> 12}

c:= a + d + b PC: Var(c) -> Unde�ned c := 14 + b

Fig. 9. Example of how dynamic rules are used to propagate constants

Figure 9 illustrates how distributing and merging of propagation rules is

accomplished. The example is decorated with two extra columns to illustrate

the result of operations on dynamic rules. For space reasons, the name of

the rule PropConst is depicted as PC. The �rst and last column represent the

input and the transformed term. The intermediate columns are introduced to

show how dynamic rules are used to propagate constant values. The second

column shows the state of the active rules. The third column shows the result

of an operation on dynamic rules which uses extra storage.

Each row in the table shows an inspected node, the result of an operation

on dynamic rules, and the transformed term. The �rst and the second row of

the table generate a propagation rule respectively. The if statement evaluates

the condition and saves active propagation rules in dr-in to be passed to its

branches. The then branch is traversed and the resulting state of propagation

rules is stored in dr-then set. To traverse the else branch with the state of

incoming propagation rules to the if statement, rules in dr-in set are restored.

At the merge point of the if statement an intersect operation is speci�ed to

obtain the active propagation rules valid after the if statement.

The speci�cation of constant propagation for an if statement is de�ned by

the following strategy:

if-cp(cp) =

If(id, id, id)

; where(save-PropConst => dr-in)

; If(id, cp, id)

; where(save-PropConst => dr-then; <restore-PropConst> dr-in)

; If(id, id, cp)

; where(<isect-PropConst> dr-then)

The strategy uses the operations save-PropConst, restore-PropConst, and

isect-PropConst to manipulate the dynamic rule. The where(s) strategy

applies a strategy s to a term without modifying the term.

In order to distribute incoming active rules to the branches of an if, the ac-

tive set of propagation rules is saved. Constant propagation continues travers-

ing the then branch, the result of this inspection is saved in the dr-then set of

dynamic rules. To proceed in the else branch, a restore operation is de�ned

to activate the dr-in set. With this state, constant propagation continues in

the else branch. To merge two di�erent versions of propagation rules from

both branches, an intersect operation is speci�ed. The intersection is speci-

�ed with the current state of dynamic rules (active rules considering the else

branch) and the dr-then set. A propagation rule is maintained in the active

set if it is de�ned in both sets and rewrites to the same constant value.

4.2.1 Generalization

The data ow schema for if statements above was speci�c for constant pro-
pagation. It turns out that the schema can be reused for other optimizations
by using di�erent save, restore, and merge operations. This generalization can
be formalized by abstracting the schema over these operations, as follows:

if(s, save, restore, merge) =

If(id, id, id)

; where(save => dr-in)

; If(id, s, id)

; where(save=> dr-then; <restore> dr-in)

; If(id, id, s)

; where(<merge> dr-then)

An example of instantiation of this schema is if(cp, save-PropConst,

restore-PropConst, isect-PropConst) for constant propagation. Di�erent

analyses and optimizations can reuse this building block by providing di�e-

rent strategies to operate on dynamic rules. In the speci�cation of the other

control-ow constructs we take this general approach into account.

while loop for loop

continuation

merge point

continuation

merge point

Fig. 10. Merging points of control ow statements.

4.3 Loop statements

Loop statements execute a number of times (including zero) the body of a loop

statement. The execution ow of a loop has a reentering path as is depicted

in Figure 10. Therefore, loop statements fork the data ow and iterate the

body of the loop.

Structured loop statements execute the iteration conditions as the �rst and

last execution. As a motivating example for constant propagation on loops;

consider the following code:

(a := 5;

b := 4;

c := 8;

d := 5;

while((c < 100)|x)

do(b := a;

e := a + b;

a := e * d + c;

c := c + e;

e := e + c;

a := b);

f := a + b + d + c + e)

)

(a := 5;

b := 4;

c := 8;

d := 5;

while((c < 100)|x)

do(b := 5;

e := 10;

a := (50 + c);

c := (c + 10);

e := (10 + c);

a := 5);

f := (15 + c + e))

In this example the variables a,b,c and d contain constant values before the

while statement is inspected. This example illustrates the following situations:

� a contains a constant value, regardless of the execution of the while state-

ment.

� b is de�ned in the body of the while, if the while is executed at least one

iteration, b contains the value of a, i.e 5, and it can be propagated outside

the while statement.

� c does not contain a constant value because it is iteratively de�ned in the

while body.

� d contains a constant value since it is not de�ned in the while.

� e contains a constant value from its �rst de�nition until its second de�nition

inside the while loop. After the while e does not contain a constant value.

while-p(s,save,restore,merge)=

While(id, id)

; where(save => dr)

; where(While(id, s))

; where(While(id, s))

; where(<merge> dr)

; While(s, s)

; where(<merge> dr)

for-p(s,save, restore,merge)=

For(id, id, id, id)

; where(save => dr)

; where(For(id, id, id, s))

; where(For(id, id, id, s))

; where(<merge> dr)

; For(id, id, id, s)

; where(<merge> dr)

Fig. 11. Data ow schemas for loops

When the assignment to f is reached, the variables a, b and d contain cons-

tant values.

The speci�cation of constant propagation has to cover all these cases. To

discover that the value of a contains a constant value, we inspect the while

body twice, in order to take into account propagation rules discovered in the

�rst inspection. This analysis considers the reentering path of the loop. After

the unfolded inspection of the while, the merge operation selects valid propa-

gation rules considering possible execution paths. With the outcome of valid

propagation rules the transformation is performed. A last merge operation is

required to prevent propagation rules coming from the while execution being

active outside the scope of the while loop. The for statement is de�ned in an

analogous way.

The de�nition of constant propagation for loop statement is shown in

Figure 11.

4.3.1 Non Pessimistic Constant Propagation

The constant propagation speci�cation so far is still pessimistic, even though

we avoid considering unreachable code. Loop statements that are not executed

are removed from the program when the condition can be evaluated. If the

analysis cannot determine if a loop will execute or not, it considers both

situations. That is the reason for the second occurrence of merge for loop

statements. This speci�cation cannot determine the value of b as constant in

the last example.

A less pessimistic approach is to keep propagation rules after a loop state-

ment when the analysis guarantees at least one execution of the body of a

loop. The is-safe strategy checks this condition and avoids performing the

second merge of propagation rules. Thus, constant propagation rules of a loop

statement are active after the loop occurrence in the program. The strategies

in Figure 12 consider these situations.

4.4 Strategies for Control-ow Statements

Now we have de�ned all ingredients for putting together a full blown constant

propagation strategy. Figure 13 combines the ingredients into a speci�cation

is-safe =

?While(Int(i),e);

where(<not(eq)> (i,0);

<not(oncetd(Break))> e)

while(s,save,restore,meet) =

While(id,id)

; where(save => dr)

; (where(While(s,id);is-safe)

; where(While(id,s))

; where(While(id,s))

; where(<meet> dr)

; While(s,s)

<+ where(While(id,s))

; where(While(id,s))

; where(<meet> dr)

; While(s,s)

; where(<meet> dr)

)

is-safe =

?For(_,Int(i),Int(t),_);

where( <geq> (t, i))

for(s,save,restore,merge) =

For(id,id,id,id)

; where(save => dr)

; where(For(id,id,id,s))

; where(For(id,id,id,s))

; where(<merge> dr)

; For(id,id,id,s)

; ( is-safe

<+ where(<merge> dr) )

Fig. 12. Non-pessimistic loop data ow schemas

const-prop2 =

bottomupS(

try(fold2(const-prop2) <+

eval-conditionals2(const-prop2))

, control-flow2

)

control-flow2(s) =

Assign(id, s)

+ IfThen(s, id)

+ If(s, id, id)

+ For(id, s, s, id)

+ While(id, id)

+ Let([try(VarDec(id,id,s))]

, id)

fold2(s) =

fold1

+ ElimIf; s

+ ElimFor

+ ElimIfThen; s

+ While(s,id); ElimWhile

eval-conditionals2(s) =

eval-conditionals(s,

assign-cp,

declVar,

save-PropConst,

restore-PropConst,

isect-PropConst)

eval-conditionals(s,

assign, declare,

save, restore, meet) =

assign

+ ifthen(s,save,restore,meet)

+ if(s,save,restore,meet)

+ for(s,save,restore,meet)

+ while(s,save,restore,meet)

+ let-exp(s,declare)

let-exp(s, declare) =

Let([try(declare)],list(s))

Fig. 13. Constant propagation for control-ow statements

of non-pessimistic intra-procedural constant propagation. The const-prop2

strategy is de�ned in terms of the bottomupS strategy and includes strate-

gies for control-ow statements. The non-pessimistic constant propagation

uses the evaluation rules ElimIf, ElimIfThen, ElimWhile, and ElimFor, to

simplify the execution ow of expressions when possible. The strategy fold2

is parameterized with the cp strategy to apply constant propagation to the

resulting paths after the simpli�cation of statements.

When constant folding reduction is not applicable we consider di�erent

execution paths as explained before. The eval-conditionals2 strategy is

de�ned in terms of assign-cp, and the data ow schemas for conditionals

and loops. The control-flow1 strategy is extended with terms that need

control when the reduction strategy is applied. The strategy control-flow2

de�nes how to traverse control-ow statements to simulate the execution of

the program.

5 Discussion

We have shown how to specify intra-procedural constant propagation using

strategic dynamic term rewriting. The techniques used in the paper such as

dynamic scoped rules and context-sensitive term traversal permitted us to

write a speci�cation which is elegant and concise.

5.1 Generalization and Future Work

As presented here, the constant propagation optimization is intra-procedural

and does not take function calls into account. We have speci�ed an extension

that analyzes function de�nitions and at function calls only unde�nes propa-

gation rules for global variables killed in the function body. Generalization

to inter-procedural constant propagation [7,5] needs further investigation, but

seems straightforward.

Other source-to-source optimizers can be built using the same building

blocks by providing di�erent propagation rules and corresponding rule oper-

ations. We have already speci�ed copy propagation, common sub-expression

elimination and dead code elimination using this approach. We are con�dent

that the techniques can be used for the speci�cation of many other machine

independent optimizations such as code motion [11], although this requires

further investigation.

The current building blocks can be easily extended to include control-ow

constructs that are not present in Tiger such as case and switch. Also exten-

sion to semi-unstructured control-ow using breaks can be dealt with. The

framework can be made reusable with little e�ort across di�erent languages.

Di�erent dynamic rules can be created to propagate or to provide infor-

mation for di�erent aspects of a program. At the moment we are interested

in investigating the possibility of combining several optimizations while a pro-

gram is being inspected in the style of [13].

Constant propagation or in general information propagation is been used

in partial evaluators [15], supercompilers [10], interpreters [8], as a few exam-

ples. In [8] dynamic rules are used to propagate information of the evaluation

state of expressions. The present work can be extended and combined with

other techniques such as program specialization and generalization in order to

construct such systems.

5.2 Previous Work

This paper is part of an ongoing investigation into the speci�cation of di�erent

kinds of program transformations using rewriting strategies and dynamic rules.

In [20] programmable rewriting strategies, in particular for generic traversal,

were introduced and applied to simpli�cation of functional programs. In [18]

dynamic rules are introduced and applied to bound variable renaming, func-

tion inlining, and dead function elimination. In [6] dynamic rules are used to

implement dynamic programming for instruction selection, i.e., associating the

lowest cost reduction with a term. The Tiger optimization presented in this

paper is part of the Tiger-in-Stratego
1
project, which is aimed at exploring

program transformation techniques for compilation.

The contribution of this paper is the formalization of merge operations

for dynamic rules and the traversal strategy that emulates the control ow

of a program. These mechanisms allow us to write the building blocks that

are reusable for other optimizations that require a forward analysis. These

building blocks can be parameterised for di�erent optimizations, providing

di�erent information through dynamic rules for each aspect that is relevant to

inspect. A backward traversal of the program was implemented using similar

constructions with dead code elimination as an instance.

5.3 Related Work

Path logic programming [9] is a related speci�cation technique for data-ow

optimizations. The applicability conditions of a transformation rule are ex-

pressed by means of regular expressions over the paths through a program.

For example, a constant assignment can be propagated to a usage site if there

are no intervening de�nitions of the variable on any path between de�nition

and use. Instead of transforming abstract syntax trees, control-ow graphs

are used as basis for transformations. The speci�cation of optimizations is

more declarative than the one presented in this paper. On the other hand,

our approach allows for the simultaneous propagation of all constants or other

properties throughout the program, whereas in the path logic programming

approach, the entire graph needs to be reanalyzed after each transformation.

1 http://www.stratego-language.org/Tiger

Expressing transformations by rewriting tends to be clear and intuitive.

There are three major factors to consider for a rewrite system. Expressiveness

of transformations, representation of control-ow paths, and data availability

to test applicability conditions to perform rewriting. Rewriting is split into

two di�erent areas: term rewriting and graph rewriting [3]. Term rewriting

expresses transformations in a clear and elegant way, but it is more diÆcult to

express applicability conditions, when data not available in the inspected term

is needed, term rewriting systems su�ers from this issue. In contrast, graph

rewriting provides mechanisms to represent the control ow of a program, the

applicability conditions of rewriting are also easier to represent, although the

rewriting itself is more diÆcult to represent [12]. Providing information to

term rewriting by means of dynamic rules and a strategy to emulate the data

ow of programs leads to a high level speci�cation for constant propagation

which is directly executable.

There are many algorithms for constant propagation, most of them use a

graph representation such as control-ow graphs with the static single assign-

ment (SSA) property [21], or even the gated single assignment property [17].

Special � assignments are introduced to provide the representation of a pro-

gram with unique variable de�nitions. The most used algorithm for constant

propagation is the Sparse conditional constant propagation (SCC) [21,17,14,2].

SCC is an optimistic algorithm which avoids to consider unreachable code and

continues the propagation of constant values in the branches of conditionals

statements. These algorithm properties are also present in our speci�cation.

Another approach uses system dependence graphs (SDG) to represent the

data-ow and control-ow of a program [5]. This representation focuses on

data dependencies of a program, which is represented by means of graphs.

This system is provided with a data-ow interpreter to propagate constant

values to the dependent nodes. Assignment nodes provide the hints to pro-

pagate constants in the graph. At each node the SDG graph is updated

following transformation rules. The SDG approach is more convenient for

source-to-source optimizers than algorithms based on control-ow graphs with

SSA property, since they do not lower the representation language. Program

reconstruction preserves the structure of the program, although building the

system dependence graph is required in order to perform the transformation.

In all publications we encountered (with [9] as exception) algorithms for

constant propagation are expressed using pseudocode or using formal English

to describe properties of a system. Our speci�cation is consice and executable

without the need for def-use chains [1] or similar analysis information. All

information required is present in the rewriting rules and the mechanisms to

operate on them suÆces for our speci�cation.

6 Conclusion

In this paper we have described the application of programmable rewriting

strategies to the speci�cation of a data-ow optimization. Programable strate-

gies allow us to emulate the data ow of a program, even though using abstract

syntax trees to represent a program. The parameterization of the traversal

strategies with the rules to apply makes it possible to de�ne di�erent opti-

mizations by providing a di�erent set of rules and even enables us to combine

di�erent optimizations.

The use of dynamic rules for the propagation of data-ow information

(instead of environments) makes it possible to separate traversal from propa-

gation. The extension of dynamic rules with operations to save, restore, and

intersect sets of rules makes it possible to model data-ow through multiple

paths. Together, these mechanisms support the combination of analysis and

transformation while the abstract syntax tree is traversed.

